

Development of a Program for the Display of Molecular Models

—MOLEX SERIES, Version I. “MOLEX1”—

By

Mitsunobu SATO

Department of Engineering, National Science Museum, Tokyo,

and

Hidehiko NAKANO

Department of Applied Chemistry, Himeji Institute of Technology, Himeji.

Abstract

A program “MOLEX1” for the computer graphics of molecular models has been developed. MOLEX1 draws ball-and-stick type molecular models with hidden line elimination using the data of Cartesian coordinates for each atom and a list of connections for the atoms. MOLEX1 requires a 64 K PC-8801 (NEC Co. Ltd.) with N₈₈-BASIC and a disk drive. Molecules which consist of up to 200 atoms can be displayed on CRT in a color version. The hard copy system of the centered projection and the stereoscopic illustration of molecular models has been also prepared.

Recently, stereochemical considerations of numerous compounds are essential for chemists. The technique of X-ray crystal structure analysis, for example, takes an important role to clarify the structures of many compounds. Generally, the geometrical structures and the absolute configurations of compounds, which have been confirmed by such a procedure, are shown using ball-and-stick type molecular models. It is important that those molecular models are available for understanding of the structures, though a little difficult procedure to draw them has been required because of the usage of computer centers.

Accompanying with recent improvement in computer technique, on the other hand, it is expected that those molecular models can be widespread. Co-worker Nakano has already developed a program “MICROMOL” (1982), which displays ball-and-stick type molecular models of centered projection on CRT and is adaptable to the 48 K Apple-II devices. In this work, we have developed a program adaptable to 64 K PC-8801 system, which has many functions for graphics. The program named “MOLEX1” consists of several subroutines. The hard copy system of the centered projection and stereoscopic illustration of molecular models has been newly added. The displays of the sticks representing bonds, moreover, have been contrived to clarify

Table 1. Input Data for Cyclohexane.

CYCLOHEXANE			
ATOM NO.	BOND NO.		
X	Y	Z	ATOM SP.
18	18		
2.33	.044	3.012	4
3.631	-.661	2.633	4
1.368	.05	1.836	4
2.525	.9	3.114	1
1.934	-.507	3.936	1
3.389	-1.675	2.469	1
4.348	-.605	3.22	1
1.031	-.972	1.607	1
.592	.581	1.921	1
3.285	-.044	.208	4
1.984	.661	.587	4
4.247	-.05	1.384	4
3.09	-.9	.106	1
3.681	.507	-.716	1
2.226	1.675	.751	1
1.267	.605	0	1
4.584	.972	1.613	1
5.023	-.581	1.299	1
connection table			
1	2		
1	3		
1	4		
1	5		
2	6		
2	7		
2	12		
3	8		
3	9		
3	11		
10	11		
10	12		
10	13		
10	14		
11	15		
11	16		
12	17		
12	18		

the respective distances of atoms.

In Table 1 and 2, input and output data for the display of cyclohexane molecule (KAHN *et al.*, 1973) are given, respectively. In Fig. 1, 2 and 3, the molecular models of cyclohexane, N-acetyl histidine methylamide (HARADA *et al.*, 1977) and diamminebis((RR)-2,2'-bipiperidine)cobalt(III) complex ion (SATO *et al.*, 1983) are shown,

Table 2. Output Data for Cyclohexane.

CYCLOHEXANE
NEW COORDINATES & NUMBERING with RADIUS

SCALING FACTORS

70.4561 -31.737

X	Y	Z	INP. NO.	R(I)
-1.16041	-.673532	2.326	5	.232481
1.85858	-.729919	1.61	7	.211134
-.336259	1.07127	1.504	4	.208302
-.561125	.0517058	1.402	1	.411296
.923977	-.74165	1.023	2	.392704
.639933	-1.84331	.859	6	.192585
-2.29125	.600864	.311	9	.180983
-1.47493	.0512308	.226	3	.358615
1.77707	.97231	3.00002E-03	17	.175056
-1.77593	-.97169	-3.00014E-03	8	.174944
1.40573	-.048827	-.226	12	.341789
2.1446	-.562407	-.311	18	.1694
-.532846	1.53485	-.859	15	.160358
-.742733	.59617	-1.023	11	.315673
.415567	-.0382931	-1.402	10	.304604
.243561	-.775947	-1.504	13	.150879
-1.31538	.51659	-1.61	16	.149427
.70034	.406494	-2.326	14	.140309

connection table

4	3
4	1
5	4
5	2
6	5
8	4
8	7
10	8
11	5
11	9
12	11
14	8
14	13
15	14
15	11
16	15
17	14
18	15

respectively. It takes about 30, 45 and 90 sec, respectively, to display each figure on CRT. In Fig. 2 (b), the molecular model of 2 (a) after rotation is given, it takes about 25 sec to calculate the new coordinates of each atoms. A stereoscopic illustration of cyclohexane is also shown in Fig. 4. On CRT, every figure is displayed in a color





Figure 1. Molecular Model of Cyclohexane.



(a)

(b)

Fig. 2. Molecular Model of N-acetyl histidine methylamide.




Fig. 3. Molecular Model of diamminebis((RR)-2,2'-bipiperidine)-cobalt (III) complex ion.




Fig. 4. A stereoscopic illustration of Cyclohexane.

version.

Though the resolution of the figures on CRT seems to be sufficient, the hard copy system may require the higher resolution of them, especially in the case of the stereoscopic illustrations. The needed time to draw the molecular models seems to be sufferable. Consequently, this "MOLEX1" system will be useful not only for chemists, but as an instrument for chemical education.

Supplementary materials (Directions for using MOLEX1 and its program list) are available.

References

- NAKANO, H., 1982. MICROMOL (Microcomputer Manipulation of Molecular Models). Copyright (c) 1982 by Hidehiko NAKANO.
- HARADA, Y. and Y. IITAKA, 1977. N-Acetyl-L-histidine-N-methylamide. *Acta Crystallogr. Sect. B.*, **33**: 250.
- KAHN, R., R. FOURME, D. ANDRE and M. RENAUD. Crystal Structure of Cyclohexane I and II. *Acta Crystallogr. Sect. B.*, **29**: 131.
- SATO, M., S. YANO and S. YOSHIKAWA, 1983. to be published.

Directions for using MOLEX1.

Two directions for using MOLEX1 have been prepared for the visitors of museums and science centers and for the instructors and researchers.

1) For visitors of museums and science centers

After the usage by a previous visitor, the message "What kind of compound do you select (1 to 5)?" will be displayed. When the visitor selects a compound which has been provided by instructors, the display of the molecular model can be started after reading the data from the diskette. As soon as the display has completed, the message "Do you rotate the molecule (Y OR N)?" will be displayed. If the answer is "N", the first message "What kind of compound do you select (1 to 5)?" will be repeated in order to choose another compound. The affirmative response gives rise to display two prompt message "Rotation Axis (X, Y, Z)?" and "Rotation Angle (Degree)?". After the visitor enters the both data, the molecular model rotated will be drawn. The prompt message "Do you rotate the molecule (Y OR N)?" will be then repeated and the corresponding picture will be shown until the negative reply will be given. It is necessary for the visitors to reply "N" when they want to terminate the execution using the current data of the selected compound.

2) For instructors and researchers

Six commands, i.e. DATA, COPY, STEREO, BOND, PLANE and END commands are available for instructors and researchers. These commands are effective when you reply "COMMAND" to the message "What kind of compound do you select (1 to 5)?", except for END command. The prompt message "PLEASE INPUT COMMAND?" will be followed. END command is available to terminate the execution of MOLEX1 when the message "What kind of compound do you select (1 to 5)?" is displayed after performance of each command repeatedly.

a) DATA command

This command should be selected when you want to store the fundamental data

of molecules with the devices.

After entering DATA, you are to type remarks, number of constituent atoms and bonds, x-, y-, z-coordinate and the specific number of each atom sequentially, according to the prompt messages displayed. Then pairs of atoms which are bonded together are to be typed in. The specific number of each atom can be arbitrarily determined from number 2 to 6. These numbers automatically decide the color of each atom on CRT, except for Hydrogen atom (blue).

As soon as the data have been typed from keyboard, sorting and transforming subroutine will be performed and then the molecular model of centered projection will be displayed. After the drawing, the prompt message "Do you input the Rotated Data on Disk (Y OR N)?" will be displayed, so new data after rotation, which is more favorable, will be able to be stored with the devices. When you reply to the inquiry for file number, these data will be written to diskette. Moreover, these data can be confirmed by output on line printer if you need.

b) COPY command

By entering COPY command, ball-and-stick type molecular model will be drawn both on CRT and line printer using the data from diskette.

c) STEREO command

By entering STEREO command, the stereoscopic illustration of the molecular models will be drawn on CRT and then on line printer using the data from diskette.

d) BOND and PLANE commands

BOND and PLANE mean that the rotation is to be performed around the axis which passes through appointed two atoms and around the axis perpendicular to the plane which is determined by appointed three atoms, respectively. These commands perform using the original data written on diskette. In any case, rotation angle should be typed in.

```

1000 /*****
1010 /   DISPLAY OF MOLECULAR MODELS
1020 /                                     * MOLEX SERIES *   Version 1.
1030 /*****
1040 /
1050 DIM X0(200),Y0(200),Z0(200),N0%(200),B0%(200),B1%(200)
1060 DIM X(200),Y(200),Z(200),R(200),TN%(200),B%(200,6),C%(200),C9%(200)
1070 WIDTH 80,25:CONSOLE 0,25,0,1
1080 N=0:M=0:PI=3.14159:F3=320:F4=100
1090 S1=SIN(PI/12):C1=COS(PI/12)
1100 PRINT :INPUT "WHAT KIND OF COMPOUND DO YOU SELECT ( 1 TO 5 )";A$
1110     IF A$="COMMAND" THEN 1180
1120     IF A$="END" THEN END
1130     GOSUB *R
1140     CLS
1150 CLS:PRINT :INPUT "DO YOU ROTATE THE MOLECULE ( Y OR N )";B$
1160     IF B$="Y" THEN GOSUB *RT
1170     CLS:GOTO 1100
1180 CLS:PRINT
1190 INPUT "PLEASE INPUT COMMAND ( DATA,COPY,STEREO,BOND,PLANE )";C$
1200     IF C$="DATA" THEN GOSUB *D
1210     IF C$="COPY" THEN GOSUB *C
1220     IF C$="STEREO" THEN GOSUB *ST
1230     IF C$="BOND" THEN GOSUB *RB
1240     IF C$="PLANE" THEN GOSUB *RP
1250     GOTO 1100
1260 /
1270 /   **** READ ****
1280 /
1290 *R
1300 PRINT:PRINT " DISK IS WORKING ! "
1310 IF A$="2" THEN 1360
1320 IF A$="3" THEN 1370
1330 IF A$="4" THEN 1380
1340 IF A$="5" THEN 1390
1350 OPEN "1:DATA1" FOR INPUT AS #1:GOTO 1400
1360 OPEN "1:DATA2" FOR INPUT AS #1:GOTO 1400
1370 OPEN "1:DATA3" FOR INPUT AS #1:GOTO 1400
1380 OPEN "1:DATA4" FOR INPUT AS #1:GOTO 1400
1390 OPEN "1:DATA5" FOR INPUT AS #1
1400 INPUT #1,E$,N,M,F1,F2
1410 FOR I=1 TO N
1420     INPUT #1,X0(I),Y0(I),Z0(I),N0%(I)
1430 NEXT
1440 FOR I=1 TO M
1450     INPUT #1,B0%(I),B1%(I)
1460 NEXT
1470 FOR I=1 TO N
1480     INPUT #1,X(I),Y(I),Z(I),C%(I),R(I)
1490 NEXT
1500 FOR I=1 TO N
1510     FOR K=0 TO 6
1520     IF EOF(1) THEN 1560
1530     INPUT #1,B%(I,K)
1540     NEXT
1550 NEXT I

```

```

1560 CLOSE #1
1570 GOSUB *P
1580 RETURN
1590 '
1600 '      ***** SORTING *****
1610 '
1620 *S
1630 CLS
1640 PRINT E$
1650 PRINT:PRINT "SORTING is RUNNING !"
1660 FOR I=1 TO N
1670   C%(I)=I:Z(I)=Z0(I)
1680 NEXT
1690 IF N=1 THEN GOTO 1840
1700 FOR I=1 TO N-1
1710   Z9=Z(I)
1720   K=I
1730   FOR J=I+1 TO N
1740     IF Z(J)<=Z9 THEN GOTO 1760
1750     Z9=Z(J):K=J
1760   NEXT
1770   Z(K)=Z(I):Z(I)=Z9
1780   C5%=C%(K):C%(K)=C%(I):C%(I)=C5%
1790 NEXT I
1800 '
1810 '      ***** TRANSFORMING & SCALING *****
1820 '
1830 PRINT:PRINT "TRANSFORMING is RUNNING !"
1840 FOR I=1 TO N
1850   X(I)=X0(C%(I)):Y(I)=Y0(C%(I))
1860 NEXT
1870 X5=0:Y5=0:Z5=0
1880 FOR I=1 TO N
1890   X5=X5+X(I):Y5=Y5+Y(I):Z5=Z5+Z(I)
1900 NEXT
1910 X5=X5/N:Y5=Y5/N:Z5=Z5/N
1920 L=0
1930 FOR I=1 TO N
1940   L1=(X(I)-X5)*(X(I)-X5)+(Y(I)-Y5)*(Y(I)-Y5)+(Z(I)-Z5)*(Z(I)-Z5)
1950   IF L1>L THEN L=L1
1960 NEXT
1970 L=SQR(L):F2=-90/(L+.3):F1=-F2*2.22:V=3*(L+.6)
1980 FOR I=1 TO N
1990   Z(I)=Z(I)-Z5:F=V/(V-Z(I))
2000   X(I)=(X(I)-X5)*F
2010   Y(I)=(Y(I)-Y5)*F
2020   R(I)=.35*F:IF N0%(C%(I))=1 THEN R(I)=R(I)/2
2030   FOR J=0 TO 6
2040     B%(I,J)=0
2050   NEXT
2060   C9%(C%(I))=I
2070 NEXT I
2080 FOR I=1 TO M
2090   B2=C9%(B0%(I)):B3=C9%(B1%(I))
2100   IF B2<B3 THEN GOTO 2140
2110   GOTO 2130

```

```

2120 IF B2>B3 THEN GOTO 2140
2130 B4=B2:B2=B3:B3=B4
2140 FOR K=0 TO 6
2150 IF B%(B3,K)>0 THEN GOTO 2170
2160 B%(B3,K)=B2:GOTO 2180
2170 NEXT
2180 NEXT I
2190 GOSUB *P
2200 RETURN
2210 '
2220 ' ***** BALL AND STICK *****
2230 '
2240 *P
2250 IF C$="STEREO" THEN 2270
2260 SCREEN 0,0,0,0:CLS 3
2270 PRINT E$:IF C$="STEREO" THEN C$="COPY"
2280 FOR I=N TO 1 STEP -1
2290 ' ***** CIRCLE *****
2300 TN%(I)=N0%(C%(I))
2310 X9=X(I):Y9=Y(I):RA=.5
2320 IF C$="COPY" THEN RA=RA*14/18.5
2330 D1=INT(X9*F1)+F3:D2=INT(Y9*F2)+F4
2340 D2=INT(Y9*F2)+F4
2350 RD=INT(R(I)*F1):IF C$="COPY" THEN RD=INT(RD*1.3)
2360 CIRCLE (D1,D2),RD,7,,,RA:PAINT (D1,D2),2,7
2370 CIRCLE (D1,D2),RD,TN%(I),,,,RA:PAINT (D1,D2),0,TN%(I)
2380 IF I=1 THEN 2670
2390 COLOR 7
2400 FOR K=0 TO 6
2410 J2=B%(I,K)
2420 IF J2=0 THEN GOTO 2660
2430 X5=X(J2):Y5=Y(J2):Z5=Z(J2)
2440 X6=X(I):Y6=Y(I):Z6=Z(I)
2450 L1=(X5-X6)*(X5-X6)+(Y5-Y6)*(Y5-Y6)
2460 L2=SQR(L1+(Z5-Z6)*(Z5-Z6))
2470 L1=SQR(L1)
2480 RJ=R(J2)*.7:RI=R(I)*.7:IF N0%(C%(I))=1 THEN RJ=R(J2)/2
2490 IF L1-RJ-RI*L1/L2<0 THEN 2660
2500 X3=X5-RJ*(X5-X6)/L1
2510 X4=X6+RI*(X5-X6)/L2
2520 Y3=Y5-RJ*(Y5-Y6)/L1
2530 Y4=Y6+RI*(Y5-Y6)/L2
2540 DX1=INT(X3*F1)+F3:DY1=INT(Y3*F2)+F4
2550 DX2=INT(X4*F1)+F3:DY2=INT(Y4*F2)+F4
2560 LINE (DX2,DY2)-(DX1,DY1)
2570 IF N0%(C%(I))=1 THEN 2640
2580 IF N0%(C%(J2))=1 THEN 2640
2590 R=1:IF DX2>DX1 THEN R=-1
2600 LINE (DX2+1*R,DY2)-(DX1+1*R,DY1+2*R)
2610 LINE (DX2+1*R,DY2)-(DX1+1*R,DY1+1*R)
2620 LINE (DX2,DY2-1*R)-(DX1-2*R,DY1-1*R)
2630 LINE (DX2,DY2-1*R)-(DX1-1*R,DY1-1*R)
2640 LINE (DX2,DY2)-(DX1+1,DY1+1)
2650 LINE (DX2,DY2)-(DX1-1,DY1-1)
2660 NEXT K
2670 NEXT I

```

```

2680 RETURN
2690 /
2700 /   ***** ROTATION *****
2710 /
2720 *RT
2730 IF C$="STEREO" THEN 2800
2740 INPUT "ROTATION AXIS ( X, Y, Z )";A$
2750 IF A$= "X" THEN 2790
2760 IF A$= "Y" THEN 2800
2770 IF A$= "Z" THEN 2810
2780 GOTO 2740
2790 U1=1:U2=0:U3=0:GOTO *U
2800 U1=0:U2=1:U3=0:GOTO *U
2810 U1=0:U2=0:U3=1:GOTO *U
2820 /
2830 /   ***** BOND ROTATION *****
2840 *RB
2850 INPUT " BOND FROM ATOM ";I
2860 INPUT "      TO ATOM ";J
2870 U1=X0(J)-X0(I):U2=Y0(J)-Y0(I):U3=Z0(J)-Z0(I)
2880 GOTO 2960
2890 /
2900 /   ***** PLANE ROTATION *****
2910 *RP
2920 INPUT "THREE ATOMS ON THE PLANE ";I,J,K
2930 U1=(Y0(J)-Y0(I))*(Z0(K)-Z0(J))-(Y0(K)-Y0(J))*(Z0(J)-Z0(I))
2940 U2=(Z0(J)-Z0(I))*(X0(K)-X0(J))-(Z0(K)-Z0(J))*(X0(J)-X0(I))
2950 U3=(X0(J)-X0(I))*(Y0(K)-Y0(J))-(X0(K)-X0(L))*(Y0(J)-Y0(I))
2960 U0=SQR(U1*U1+U2*U2+U3*U3)
2970 IF U0>.01 THEN 3000
2980 PRINT "IRREGULAR COMBINATION ! ":PRINT " PLEASE REINPUT."
2990 GOTO 2740
3000 U1=U1/U0:U2=U2/U0:U3=U3/U0
3010 *U:U0=SQR(U1*U1+U2*U2)
3020 IF U0<.5 THEN 3040
3030 U4=U2/U0:U5=-U1/U0:U6=0:GOTO 3060
3040 U0=SQR(U1*U1+U3*U3)
3050 U4=U3/U0:U5=0:U6=-U1/U0
3060 U7=U3*U5-U2*U6:U8=U1*U6-U3*U4:U9=U2*U4-U1*U5
3070 IF C$="STEREO" THEN 3090
3080 INPUT "ROTATION ANGLE";T
3090 S1=SIN(T*PI/180):C1=COS(T*PI/180)
3100 FOR I=1 TO N
3110 X3=X0(I):Y3=Y0(I):Z3=Z0(I)
3120 X4=U7*X3+U8*Y3+U9*Z3
3130 Y4=U4*X3+U5*Y3+U6*Z3
3140 Z4=U1*X3+U2*Y3+U3*Z3
3150 X5=X4*C1-Y4*S1
3160 Y5=X4*S1+Y4*C1
3170 X0(I)=U7*X5+U4*Y5+U1*Z4
3180 Y0(I)=U8*X5+U5*Y5+U2*Z4
3190 Z0(I)=U9*X5+U6*Y5+U3*Z4
3200 NEXT I
3210 GOSUB *S
3220     IF W$="Y" THEN RETURN 3940
3230     RETURN 1150

```

```

3240 /
3250 /      ***** STEREO VIEW *****
3260 *ST
3270 INPUT " INPUT FILE NO. ";A$
3280   CLS 3
3290   WINDOW (0,0)-(640,200)
3300   VIEW (0,40)-(400,165)
3310   T=-1.5:GOSUB *RT
3320   C$="STEREO"
3330   WINDOW (0,0)-(640,200)
3340   VIEW (240,40)-(639,165)
3350   T=1.5:GOSUB *RT
3360   C$="STEREO"
3370   GOSUB *C
3380 CLS:SCREEN 0,0,0,0
3390 RETURN
3400 /
3410 /      ***** HARD COPY *****
3420 *C
3430 IF C$="STEREO" THEN 3480
3440 INPUT " INPUT FILE NO. ";A$
3450   CLS 3:PRINT:PRINT "PLEASE SET LP ! HURRY !!"
3460   IF C$="STEREO" THEN 3480
3470   GOSUB *R
3480   COPY 2:C$=" ":LPRINT CHR$(27);"H"
3490 RETURN
3500 /
3510 /      ***** DISK I/O *****
3520 /
3530 *D
3540 CLS:INPUT "DO YOU EDIT NEW OR OLD DATA ( N OR O )";ED$
3550 CLS:IF ED$="O" THEN INPUT "FILE NO.";A$:GOSUB *R
3560 IF ED$="O" THEN GOTO 3670
3570 CLS:INPUT "REMARKS";E$:PRINT
3580 INPUT "NO. OF ATOMS";N:PRINT:INPUT "NO. OF BONDS";M:PRINT
3590 PRINT " X, Y, Z, ATOMIC NO."
3600 FOR I=1 TO N
3610   PRINT I;:INPUT X0(I),Y0(I),Z0(I),N0%(I)
3620 NEXT
3630 FOR I=1 TO M
3640   PRINT I;:INPUT "BOND FROM a TO b";B0%(I),B1%(I)
3650 NEXT
3660 GOSUB *S
3670 INPUT "DO YOU CORRECT THE INPUT DATA ( Y OR N )";EA$
3680 IF EA$="N" THEN GOSUB *S
3690 IF EA$="N" THEN GOTO 3940
3700 CLS:INPUT "INPUT CORRECT DATA SYMBOL ( N,M,X,Y,Z,SP,B )";EE$
3710 IF EE$="N" THEN INPUT "NEW NO. OF ATOMS";NN:NX=NN-N:N=NN:GOTO 3830
3720 IF EE$="M" THEN INPUT "NEW NO. OF BONDS";MM:MX=MM-M:M=MM:GOTO 3890
3730 INPUT "INPUT CORRECT DATA NO. or ORD. OF CONNCT. TABLE";I
3740 IF EE$="B" THEN GOTO 3800
3750 PRINT "OLD DATA OF THE ATOM":PRINT X0(I),Y0(I),Z0(I),N0%(I)
3760 IF EE$="X" THEN INPUT "INPUT NEW DATA OF X";X0(I)
3770 IF EE$="Y" THEN INPUT "INPUT NEW DATA OF Y";Y0(I)
3780 IF EE$="Z" THEN INPUT "INPUT NEW DATA OF Z";Z0(I)
3790 IF EE$="SP" THEN INPUT "INPUT NEW NO. OF ATOM SPECIES";N0%(I)

```

```

3800 IF EE$="B" THEN INPUT "INPUT NEW PAIR OF BONDS";B0%(I),B1%(I)
3810 IF EE$="B" THEN GOTO 3670
3820 PRINT "NEW DATA OF THE ATOM":PRINT X0(I),Y0(I),Z0(I),N0%(I)
3830 FOR P=NX TO 1 STEP -1
3840 INPUT "INPUT NEW DATA NO.";I
3850 PRINT "OLD DATA OF THE ATOM":PRINT X0(I),Y0(I),Z0(I),N0%(I)
3860 PRINT:INPUT "INPUT NEW DATA OF THE ATOM";X0(I),Y0(I),Z0(I),N0%(I)
3870 NEXT P
3880 CLS:GOTO 3670
3890 FOR Q=MX TO 1 STEP -1
3900 INPUT "ORD. OF CONNECTION TABLE";I
3910 INPUT "INPUT NEW PAIR OF BONDS";B0%(I),B1%(I)
3920 NEXT Q
3930 CLS:GOTO 3670
3940 PRINT:INPUT "DO YOU INPUT THE ROTATED DATA ON DISK ( Y OR N )";W$
3950 IF W$="N" THEN 3970
3960 GOSUB *RT:GOTO 3940
3970 INPUT "DATA FILE NO.";A$
3980 IF A$="2" THEN 4030
3990 IF A$="3" THEN 4040
4000 IF A$="4" THEN 4050
4010 IF A$="5" THEN 4060
4020 OPEN "DATA1" FOR OUTPUT AS #1:GOTO 4070
4030 OPEN "DATA2" FOR OUTPUT AS #1:GOTO 4070
4040 OPEN "DATA3" FOR OUTPUT AS #1:GOTO 4070
4050 OPEN "DATA4" FOR OUTPUT AS #1:GOTO 4070
4060 OPEN "DATA5" FOR OUTPUT AS #1
4070 WRITE #1,E$;N;M;F1;F2
4080 FOR I=1 TO N
4090 WRITE #1,X0(I);Y0(I);Z0(I);N0%(I)
4100 NEXT
4110 FOR I=1 TO M
4120 WRITE #1,B0%(I);B1%(I)
4130 NEXT
4140 CLOSE #1
4150 IF A$="2" THEN 4200
4160 IF A$="3" THEN 4210
4170 IF A$="4" THEN 4220
4180 IF A$="5" THEN 4230
4190 OPEN "1:DATA1" FOR APPEND AS #1:GOTO 4240
4200 OPEN "1:DATA2" FOR APPEND AS #1:GOTO 4240
4210 OPEN "1:DATA3" FOR APPEND AS #1:GOTO 4240
4220 OPEN "1:DATA4" FOR APPEND AS #1:GOTO 4240
4230 OPEN "1:DATA5" FOR APPEND AS #1
4240 FOR I=1 TO N
4250 WRITE #1,X(I);Y(I);Z(I);C%(I);R(I)
4260 NEXT
4270 FOR I=1 TO N
4280 FOR K=0 TO 6
4290 WRITE #1,B%(I,K)
4300 NEXT
4310 NEXT I
4320 CLOSE #1
4330 INPUT " LP READY ( Y OR N )";Q$
4340 IF Q$="Y" THEN 4370
4350 INPUT " DO YOU PRINT THE DATA ON LP ( Y OR N )";P$

```

```

4360 IF P#="Y" THEN 4330 ELSE 4850
4370 IF A#="2" THEN 4420
4380 IF A#="3" THEN 4430
4390 IF A#="4" THEN 4440
4400 IF A#="5" THEN 4450
4410 OPEN "DATA1" FOR INPUT AS #1:GOTO 4460
4420 OPEN "DATA2" FOR INPUT AS #1:GOTO 4460
4430 OPEN "DATA3" FOR INPUT AS #1:GOTO 4460
4440 OPEN "DATA4" FOR INPUT AS #1:GOTO 4460
4450 OPEN "DATA5" FOR INPUT AS #1
4460 INPUT #1,E#,N,M,F1,F2
4470 PRINT E#,N,M
4480 LPRINT E#:LPRINT
4490 LPRINT "ATOM NO.          BOND NO.          ":LPRINT
4500 LPRINT N,M:LPRINT
4510 LPRINT " X              Y              Z              ATOM SP.":LPRINT
4520 FOR I=1 TO N
4530 INPUT #1,X0(I),Y0(I),Z0(I),N0%(I)
4540 PRINT X0(I),Y0(I),Z0(I),N0%(I)
4550 LPRINT X0(I),Y0(I),Z0(I),N0%(I)
4560 NEXT
4570 LPRINT:LPRINT " connection table":LPRINT
4580 FOR I=1 TO M
4590 INPUT #1,B0%(I),B1%(I)
4600 PRINT B0%(I),B1%(I)
4610 LPRINT B0%(I),B1%(I)
4620 NEXT
4630 IF P#="N" THEN 4690
4640 FOR I=1 TO 3:LPRINT:NEXT:LPRINT E#
4650 LPRINT " NEW COORDINATES & NUMBERING with RADIUS":LPRINT
4660 LPRINT " SCALING FACTORS ":LPRINT:LPRINT F1,F2:LPRINT
4670 LPRINT " X              Y              Z              INP. NO.          R(I) "
4680 LPRINT
4690 FOR I=1 TO N
4700 INPUT #1,X(I),Y(I),Z(I),C%(I),R(I)
4710 PRINT X(I),Y(I),Z(I),C%(I),R(I)
4720 LPRINT X(I),Y(I),Z(I),C%(I),R(I)
4730 NEXT
4740 LPRINT:LPRINT " connection table ":LPRINT
4750 FOR I=1 TO N
4760 FOR K=0 TO 6
4770 IF EOF(1) THEN 4840
4780 INPUT #1,B%(I,K)
4790 PRINT I,B%(I,K)
4800 IF B%(I,K)=0 THEN 4820
4810 LPRINT I,B%(I,K)
4820 NEXT
4830 NEXT I
4840 CLOSE #1
4850 C#=" "
4860 RETURN

```